

内 容 简 介

本书是为物理化学教学和科研而编写的 BASIC 语言程序集。机型以 PC-1500 为主,略加修改即可用于其它机种,例如 PB-700, MZ-731 等。全书共两篇。第一篇包括四章,涉及物理化学教学中的主要分支学科。这些程序可进行数值计算、数据处理、实验结果的统计分析和试验设计。第二篇包括两章,介绍“QN 系列多功能一般数据处理通用程序”和“一元函数模型全自动识别系列”。前者提供数值计算、制表和绘图的通用程序,后者能为实验工作者快速寻找符合实验曲线的解析函数形式带来极大方便。

本书可供高等学校化学系和化工系的高年级学生、研究生和教师以及物理化学及有关学科的科学工作者参考。

微机在物理化学中的应用

BASIC 语言程序集

王作新 赵传钧 编著
阎泽群 潘强余
责任编辑 白明珠

科学出版社出版

北京朝阳门内大街 137 号

中国科学院印刷厂印刷

新华书店北京发行所发行 各地新华书店经售

· *

1988 年 2 月第 一 版 开本: 787×1092 1/32

1988 年 2 月第一次印刷 印张: 20 1/2

印数: 0001—4,250 字数: 467,000

ISBN 7-03-000041-2/O·8

定价: 5.20 元

前 言

数值计算和数据处理在教学、科研和管理工作中的重要性是众所周知的。在物理化学中也是如此。

随着各种微型和袖珍型电子计算机的广泛使用，如何充分发挥计算机的各种功能，提高数值计算和数据处理的质量及速度，使广大物理化学科技工作者从烦琐的人工计算、制表和绘图作业中解放出来，已成为当务之急。为此，在我们自己科研和教学实践的基础上编写了这本《微机在物理化学中的应用》。

本书第一篇“物理化学常用程序汇编”中的 59 个程序涉及物理化学的各个分支，包括热力学和统计热力学、反应动力学、电化学、结构化学和胶体化学等。利用这些程序不仅可以进行一般的数值计算和实验数据处理，还能进行实验结果的统计分析和试验设计。由于程序的难易程度不一，所以编写时注意到选用从简单到复杂的各种常用计算方法。可供大学生、研究生、教师和科技工作者等选用。

本书第二篇“QN 系列多功能一般数据处理通用程序”和“一元函数模型全自动识别系列”实际上是一个独立的部分。它不仅可以在各种物理化学计算中调用，也完全可以在其它各行各业的科研、教学以至管理工作中独立运用。

我们提供的 BASIC 语言程序虽然是基于 PC-1500 袖珍计算机的，但略加修改即可在其它机种如 PB-700, MZ-731, Apple II 和 IBM PC-XT 等上使用。QN 系列多功能一般数据处理通用程序在 PC-1500, PB-700 和 MZ-731 机上的

商品软件也已研制成功。

由于我们的水平有限,书中错误和不足之处在所难免,敬请各界批评指正。

编著者

1986年7月

目 录

第一篇 物理化学常用程序汇编

第一章 热力学和统计热力学.....	1
§ 1-1 van der Waals 方程计算实际气体摩尔体积	1
§ 1-2 纯气体的逸度	5
§ 1-3 Kirhhoff 定律的应用.....	10
§ 1-4 燃烧热的测定	15
§ 1-5 最高火焰温度	40
§ 1-6 量热熵	49
§ 1-7 热力学函数一次偏导数的变换关系	65
§ 1-8 偏摩尔体积	74
§ 1-9 光谱熵	79
§ 1-10 自由能函数 $\frac{G^{\circ} - E_0^{\circ}}{T}$ 和焓函数 $\frac{H^{\circ} - E_0^{\circ}}{T}$ 的计算 ...	85
§ 1-11 Clausius-Clapeyron 方程的应用	94
§ 1-12 液体饱和蒸气压的测定.....	108
§ 1-13 二元气液相图.....	113
§ 1-14 二元理想溶液泡点计算.....	123
§ 1-15 实际二元溶液泡点计算.....	128
§ 1-16 求化学平衡与相平衡体系中的独立组分数和独立方程.....	137
§ 1-17 从标准生成热和标准熵计算平衡常数.....	146
§ 1-18 从自由能函数求平衡常数.....	152
§ 1-19 压力对平衡的影响.....	158
§ 1-20 氨基甲酸铵分解反应平衡常数的测定.....	164
§ 1-21 酸-碱平衡或络合平衡中复杂函数的计算	169

第二章 反应动力学	182
§ 2-1 反应级数的确定(微分法)	182
§ 2-2 反应级数的确定(积分法)	188
§ 2-3 反应级数的确定(半衰期法)	194
§ 2-4 乙酸乙酯皂化反应动力学测定	201
§ 2-5 蔗糖转化反应速率常数的测定	205
§ 2-6 甲酸氧化动力学	210
§ 2-7 连串反应动力学的计算	216
§ 2-8 时钟反应	225
§ 2-9 Michaelis-Menton 方程参数的确定	234
§ 2-10 有抑制剂存在的酶催化反应的类型鉴别和动力学参数的确定	237
§ 2-11 多相催化反应动力学参数的确定	246
§ 2-12 Arrhenius 方程求活化能	258
§ 2-13 H_2O_2 分解动力学测定	260
§ 2-14 消泡动力学测定	267
第三章 电化学、胶体化学及其它	273
§ 3-1 无限稀释摩尔电导	273
§ 3-2 离子平均活度系数	278
§ 3-3 原电池热力学	284
§ 3-4 吸附等温式(1)	289
§ 3-5 吸附等温式(2)	294
§ 3-6 吸附等温式(3)	305
§ 3-7 沉降系数	311
§ 3-8 高聚物的分子量	314
§ 3-9 溶液中的等温吸附测定	321
§ 3-10 由接触角计算高聚物的表面张力	326
§ 3-11 两平均值的比较	334
§ 3-12 平均值与标准值的比较	338
§ 3-13 成对比较实验结果的统计分析	342
§ 3-14 试验的区组设计——一种方式分组	346

§ 3-15 区组设计——两种方式分组	351
§ 3-16 正交设计	357
第四章 结构化学	367
§ 4-1 确定分子点群的程序	367
§ 4-2 原子轨道和原子谱项在各种对称环境中的分裂结果	371
§ 4-3 辅助识别高聚物红外谱图的程序	377
§ 4-4 分子结构-性能的多元线性回归	389
§ 4-5 径向分布函数曲线	401
§ 4-6 原子轨道中电子的几率密度等值面图	408
§ 4-7 分子中原子的自然坐标变换成直角坐标	416
§ 4-8 简单分子轨道(HMO) 法	425
参考文献	444

第二篇 QN 系列多功能一般数据处理通用程序

第五章 QN 的功能、使用方法和程序	448
§ 5-1 内容概要,主程序	448
§ 5-2 QN 系列程序的特点	451
§ 5-3 QN 系列	452
§ 5-4 QN-AU 一元函数模型全自动识别系列	527
第六章 公用服务子程序	605
§ 6-1 一元线性回归子程序 1000	605
§ 6-2 多元线性回归子程序 1050	607
§ 6-3 计算结果输出子程序 5000	610
§ 6-4 求取数据范围子程序 4254	611
§ 6-5 确定数据数量级,调整数据范围,定坐标刻度值子程序 4310	612
§ 6-6 绘图子程序块	615
§ 6-7 打印数据表格子程序块	627
§ 6-8 调用服务子程序编制专用程序实例	635
参考文献	642

第一章 热力学和统计热力学

§ 1-1 van der Waals 方程计算实际气体摩尔体积

1-1-1 原理^[1]

van der Waals 方程可写成如下形式:

$$f(\tilde{V}) = \tilde{V}^3 - \left(b + \frac{RT}{p}\right) \tilde{V}^2 + \frac{a}{p} \tilde{V} - \frac{ab}{p} = 0 \quad (1)$$

式中 $\tilde{V}$ 是气体的摩尔体积 (L mol^{-1}), p 是压力 (atm). van der Waals 常数 a 和 b 的单位分别取 $\text{atm L}^2 \text{mol}^{-2}$ 和 L mol^{-1} . T 是热力学温度 (K), 气体常数 $R = 0.08206 \text{ L atm K}^{-1} \text{mol}^{-1}$.

(1) 式是 $\tilde{V}$ 的实系数三次方程. 当 a, b, T 和 p 已知时, 解此三次方程即可得气体摩尔体积 $\tilde{V}$.

1-1-2 算法

按照理想气体状态方程算出的摩尔体积 V_0 为初值, 用 Newton-Raphson 法迭代求出(1)式的实根即为所求气体在指定条件下的摩尔体积. 因有时迭代会出现不收敛的情况, 故指定最大迭代次数为 50. 当迭代次数超过 50 时就作为不收敛处理^[2].

1-1-3 变量

A\$	存放物质名的字符串变量
A	van der Waals 常数 a ($\text{atm L}^2 \text{mol}^{-2}$)
B	van der Waals 常数 b (L mol^{-1})
N	要计算的数据点数

P(N)	压力 (atm)
T(N)	热力学温度 (K)
V(N)	摩尔体积 (L mol^{-1})
VO	初值 V_0
F	$f(\tilde{V})$ 之值
DF	$f'(\tilde{V})$
IM	最大迭代次数(=50)
I	实际迭代次数
VV	工作单元

1-1-4 程序

```

10:REM get U in U
    AN DER WAALS e
    quation
20:"U":CLEAR
30:DIM P(20),U(20
    ),T(20)
40:INPUT "substan
    ce name=";A$
50:INPUT "a=";A
60:INPUT "b=";B
70:IM=50
80:INPUT "data po
    int number=";N

90:LPRINT :LPRINT
    A$
100:LPRINT "a=";A
110:LPRINT "b=";B
120:CSIZE 1
130:LPRINT TAB 3;"
    T/K";TAB 13;"P
    /atm";TAB 26;"
    U/(l/mol)"

```

```

140:FOR J=1TO N
150:L=J-1
160:I=0
170:INPUT "t(c)=";
      T(L)
180:INPUT "P(atm)="
      ";P(L)
190:T(L)=T(L)+273.
      15
200:UU=0.08206*T(L
      )/P(L)
210:F=(P(L)+A/UU^2
      )*(UU-B)-0.082
      06*T(L)
220:IF ABS (F)<=1.
      0E-08THEN GOTO
      280
230:DF=(-2.0*A/UU^
      3)* (UU-B)+P(L)
      +A/UU^2
235:IF DF=0THEN DF
      =DF+1.0E-5
240:UU=UU-F/DF
250:I=I+1
260:IF I>=IMTHEN
      GOTO 310
270:GOTO 210
280:U(L)=UU
290:LPRINT USING "
      ####.##";T(L);
      TAB 11;USING "
      ####.###";P(L)
      ;TAB 24;USING
      "####.####";U(
      L)
295:PRINT "U=";U(L
      );"(l/mol)"
300:GOTO 320

```

```

310:LPRINT USING "
      ####.##";T(L);
      TAB 11;USING "
      ####.###";P(L)
      ;TAB 27;"-----
      "

```

```

315:PRINT "NO CONVERGE"

```

```

320:NEXT J

```

```

330:CSIZE 2:USING

```

```

340:END

```

1-1-5 用法

启动程序后依次出现下列显示,相应地键入数据如下:

显 示	键入内容
substance name =	物质名(英文名或分子式)
a =	van der Waals 常数 a
b =	van der Waals 常数 b
data point number =	要计算的数据点数

完成上述操作后,打印机打出物质名, van der Waals 常数 a 和 b 的值,以及下列形式的表头:

$T/K \quad p/\text{atm} \quad V/L \text{ mol}^{-1}$

显示 " $t(C) =$ ", 键入第一个温度值 ($^{\circ}\text{C}$). 显示 " $P(\text{atm}) =$ ", 键入相应的压力值 (atm). 计算后若收敛, 则打印出 T , p 和 V 值, 并显示 " $V = \text{####. #### (L mol}^{-1}\text{)}$ ". 若不收敛, 则打印出 T , P 值和 "—" 符号, 并显示 "NO CONVERGS". 掀 ENTER 键, 又显示 " $t(C) =$ ", 键入第二个温度值, 以后各步和上述相同, 进行第二个数据点的计算. 直到算完最后一个数据点, 掀 ENTER 键就显示提示符 $>$.

1-1-6 实例^[2]

丙烷的 van der Waals 常数 $a = 8.664 \text{ atm L}^2 \text{ mol}^{-2}$, $b = 0.08445 \text{ L mol}^{-1}$. 计算 0°C , 10 atm ; -20°C , 5 atm ; -47.69°C , 5 atm , 10 atm 和 20 atm 时的摩尔体积。

按次序键入 C_3H_8 , 8.664, 0.08445 和 5 后即打印出物质名 (C_3H_8), a 值, b 值和表头。然后依次键入各组温度和压力值, 得到的打印结果如下:

C_3H_8

$a = 8.664$

$b = 0.08445$

T/K	p/atm	$V/\text{L mol}^{-1}$
273.15	10.000	1.8874
253.15	5.000	3.7924
225.46	5.000	3.2681
225.46	10.000	1.3203
225.46	20.000	—

§ 1-2 纯气体的逸度

1-2-1 原理^[1]

纯气体的逸度系数由下式计算

$$\ln \phi = \ln \frac{f}{p} = - \frac{1}{RT} \int_0^p \alpha dp \quad (1)$$

式中 ϕ 是逸度系数, f 是纯气体的逸度, p 是气体压力, 则

$$\alpha = RT/p - \tilde{V},$$

$\tilde{V}$ 是气体摩尔体积。

纯气体在压力 p 和温度 T 时的逸度为

$$f = \phi p \quad (2)$$

1-2-2 算法

实验测得温度 T 时一系列压力下的摩尔体积 $\tilde{V}$ 值, 先进行插值, 然后用 Simpson 数值积分法^[3]求 (1) 式右方积分的数值。进而计算逸度系数和逸度。

积分步长逐渐减少, 直到减少步长算出的积分值变化不大(例如, 不大于 1%) 为止。由于 PC-1500 型计算机对数组的限制, 故步长减小到使插值点超过 200 时也输出。

1-2-3 变量

A\$	气体名称
N	输入的实验数据组数
X(N)	压力 (atm)
Y(N)	体积 $\tilde{V}(\text{L mol}^{-1})$
Ep	要求数值积分计算达到的精度 ($= 0.01$)
L	积分时分割的区间数
T	温度 (K)
A	积分下限 ($= 0 \text{ atm}$)
B	积分上限 (atm), 即要求的逸度所对应的压力
E(L)	积分自变量的值(即压力值)
F(L)	积分依变量的值(即 α/RT 的值)
S	积分 $\int_0^p \frac{\alpha}{RT} dp$ 的值
FI	逸度系数 ϕ
FF	逸度 $f(\text{atm})$

1-2-4 程序

10: CLEAR

```

20: DIM X(30), Y(30
    ), E(200), F(200
    )
30: RESTORE 310:
    READ A$
40: READ N, L, EP, T
50: COLOR 0: CSIZE
    2
55: LPRINT A$: LF 1
60: LPRINT "T="; T;
    "K"
70: LF 1
75: S0=0
80: FOR K=1 TO N
90: READ X(K), Y(K)
100: Y(K)=1/X(K)-Y(
    K)/0.08205/T
110: NEXT K
120: A=0: L0=L
130: INPUT "P(atm)=
    "; B
140: IF B=0 GOTO 250
150: GOSUB 570
160: FI=EXP (-S)
170: FF=FI*B
180: LPRINT "P=";
    USING "#####.#"
    "; B; " atm"
185: LPRINT "f/P=";
    USING "##.###"
    ; FI
190: LPRINT "f=";
    USING "#####.#"
    "; FF; " atm"
200: USING : LF 1
210: L=L0
220: GOTO 130
250: END

```

310: DATA "NH3", 8, 2
 , 1E-2, 473.15

320: DATA 20, 1.866,
 60, .5708, 100, .
 3109, 150, .1767
 , 200, .1074, 250
 , .07418

330: DATA 300, .0596
 , 400, .04768

400: I=1

410: IF T<0.5*(X(I+
 1)+X(I+2)) THEN
 GOTO 460

420: IF T>=0.5*(X(N
 -2)+X(N-1))
 THEN GOTO 450

430: I=I+1

440: GOTO 410

450: I=N-2

460: M=I+2

470: F=0.0

480: FOR J=I TO M

490: P=1.0

500: FOR K=I TO M

510: IF J-K=0 THEN
 GOTO 530

520: P=P*(T-X(K))/(
 X(J)-X(K))

530: NEXT K

540: F=F+P*X(J)

550: NEXT J

560: RETURN

570: S0=0

580: H=(B-A)/2/L

590: GOSUB 800

600: GOSUB 700

610: IF ABS (S-S0)<
 =EP*ABS (S)

输入数据

插值子程序

```

        GOTO 670
    620: S0=S
    640: L=2*L
    645: IF 2*L+1>200
        GOTO 670
    650: GOTO 580
    670: RETURN
    700: S=F(2*L+1)-F(1
        )
    710: FOR W=1 TO L
    720: S=S+2*F(2*W-1)
        +4*F(2*W)
    730: NEXT W
    740: S=S*H/3
    750: RETURN
    800: T=A
    820: FOR Z=1 TO 2*L+
        1
    830: GOSUB 400
    840: F(Z)=F: E(Z)=T
    850: T=T+H
    860: NEXT Z
    870: RETURN

```

Simpson 积分子程序

1-2-5 用法

在 310 以后的 DATA 语句中依次存放 A\$, N, L, Ep, T 以及各组 $p, \tilde{V}$ 值. 注意: $p, \tilde{V}$ 值要按 P 从小到大依次存放.

L 一般取一个小的整数 (如 2 或 3), 以后在变步长过程中将自动成倍增长.

启动程序后按显示内容键入所要计算的压力值 (atm). 打印输出: 物质名, 温度 (K), 压力 (atm), 逸度系数和逸度 (atm).

1-2-6 实例

利用下列数据计算 200℃ 下, 压力分别等于 50, 100 和 400 atm 时 NH_3 的逸度:

p/atm	20	60	100	150	200	250	300	400
$\tilde{V}/\text{ml mol}^{-1}$	1866	570.8	310.9	176.7	107.4	74.18	59.60	47.68

数据输入格式见程序 310—330, 注意: $\tilde{V}$ 的单位已转换成 L mol^{-1} .

打印输出:

NH_3

$T = 473.15\text{K}$

$P = 50.0 \text{ atm}$

$f/P = 0.907$

$f = 45.3 \text{ atm}$

$p = 100.0 \text{ atm}$

$f/p = 0.822$

$f = 82.2 \text{ atm}$

$p = 400.0 \text{ atm}$

$f/p = 0.462$

$f = 184.9 \text{ atm}$

§ 1-3 Kirkhoff 定律的应用

1-3-1 原理

化学反应

$$\sum_{i=1}^m \nu_i R_i = 0 \quad (1)$$

式中 m 是反应体系的物种数, ν_i 是组分 i 的化学计量系数, R_i 是组分 i 的化学式。

设组分 i 的恒压摩尔热容可用下列通式表示

$$\tilde{C}_{p,i} = a_i + b_i \times 10^{-3}T + c_i \times 10^{-6}T^2 + d_i \times 10^5T^{-2} \quad (2)$$

$\tilde{C}_{p,i}$ 的单位是 $\text{J K}^{-1}\text{mol}^{-1}$, T 的单位是 K 。则

$$\begin{aligned} \Delta C_p &= \sum_i \nu_i \tilde{C}_{p,i} \\ &= \sum_i \nu_i a_i \\ &\quad + \sum_i \nu_i b_i \times 10^3 T + \sum_i \nu_i c_i \times 10^{-6} T^2 \\ &\quad + \sum_i \nu_i d_i \times 10^5 T^{-2} \end{aligned} \quad (3)$$

已知温度 T_1 的反应热 ΔH_1 , 利用 Kirhhoff 定律得出计算温度 T_2 的反应热 ΔH_2 的公式为

$$\begin{aligned} \Delta H_2 &= \Delta H_1 + \sum_i \nu_i a_i (T_2 - T_1) \\ &\quad + \frac{1}{2} \sum_i \nu_i b_i \times 10^{-3} (T_2^2 - T_1^2) \\ &\quad + \frac{1}{3} \sum_i \nu_i c_i \times 10^{-6} (T_2^3 - T_1^3) \\ &\quad - \sum_i \nu_i d_i \times 10^5 \left(\frac{1}{T_2} - \frac{1}{T_1} \right) \end{aligned} \quad (4)$$

1-3-2 算法

只要已知各物质的 $\tilde{C}_{p,i}$ 式中的常数值和任意一个温度下的反应热, 直接由 (4) 式计算另一温度下的反应热。